

ABC ADC

A B C A D C 3 2 M A N U A L

***** INNEHÅLL *****

Sida

- 1 Datablad
- 2 Tekniska data
- 3 Beskrivning
- 4 Stiftsplacering
- 5 Kortval, Förstärkning, Tider
- 6 Koppling +- 10V, +- 5V
- 7 Koppling OV - 10V
- 8 Kommandon, BASIC och assembler
- 9 Programexempel BASIC
- 10 Programexempel assembler
- 11 Inkoppling av analoga signaler
- 12 Metoder för störundertryckning
- 13 Färdiga drivprogram
- 14 Installation
- 15 Placeringsschema
- 16 Kopplingsschema

SCANDIA METRIC AB

BANVAKTSV. 20, FACK, 17119 SOLNA, TEL 08/820400
DANMARK TEL 02/804200 NORGE TEL 02/282624 FINLAND TEL 90/460844

ABC ADC


ABC ADC32 är ett anpassningskort för mätning av analoga signaler med ABC 80. Detta kort har stort användningsområde, då man ofta vill ansluta analoga mätsignaler direkt till datorn för vidare bearbetning. ABC ADC32 har 32 separata ingångar (single ended) eller 16 differentiella eller en valfri kombination av dessa ingångar.

ABC ADC32 uppfyller många krav som kan ställas vid mätning av analoga signaler, såsom valfri förstärkning, single ended/differentiell mätning, snabb omvandling och en digital ingång för exv. start av mätsekvens. För att förstärka inkommende signal har ABC ADC32 en fast och en programstyrd förstärkare före själva analog till digital-omvandlingen. Den analoga insignalen till ABC ADC32 kan ställas från +- 10 Volt till +- 40 millivolt, fullt utslag.

----- ANVÄNDNINGSSOMRÅDE -----

ABC ADC32 kan användas i många typer av analog mätning såsom:

- == Mätinstrument med analog utgång
- == Förstärkta signaler från:
 - == Temperaturgivare
 - == Tryckgivare
 - == Trådtöjningsgivare
 - == Kraftgivare
 - == Flödesmätare



----- UPPBYGGNAD -----

ABC ADC32 är uppbyggd kring en 12-bits analog till digital-omvandlare. Varje kanal väljs med multiplexer (MUX). Signalen förstärks i en differentialförstärkare och i en programmerbar förstärkare. Sample/hold-kretsen (S/H) håller signalen stabil medan A-D-omvandlaren konverterar signalen till motsvarande digitala värde. Inställning av differentiell eller single ended mätning, kanal, eventuell förstärkning samt start av konvertering kontrolleras helt från datorn. En digital bit in från analogkontakten kan användas för exv. triggning.

----- GENERELLA DRIVRUTINER -----

Till ABC ADC32 finns färdiga drivrutiner för att enkelt klara av omfattande mätningar. Drivrutinerna används för att få exakt tidsintervall, snabb avläsning och lättåtkomligt resultat. Hanteringen av drivrutinerna sker helt i BASIC.

==== ALLMÄNT ===

Spänningssmatning	:	+ 5V
Strömförsljning	:	400 mA
Intern spänning	:	DC/DC för +-15 V för de analoga komponenterna
Storlek	:	Europakort 100x160 mm
Bussanslutning	:	Till ABC-80-bussen
Kortadresser	:	6 byglingar ger 64 möjliga adresser
Kontaktdon	:	96-pin Europadon, DIN 41612
Temperaturområde	:	0 - 70 grader Celsius
Avläsningstid	:	Min 2 ms i BASIC Min 60 µs i maskinspråk

==== INGÅNGAR ===

Antal kanaler	:	32 vid koppling med gemensam jord
	:	16 vid differenziell koppling
Ingångsimpedans	:	100 Mohm
Överspänning	:	Max +- 20 V
Digital ingång	:	En extern bit för programstyrning o. dyl.

==== FÖRSTÄRKNING, MÄTOMRÅDE ===

Fast i Diff.först.	:	1, 10, 100 ggr
Programmerbar	:	1, 2, 4, 8, 16, 32, 64, 128 ggr
Ingångsnivåer till	:	0 till 10 V
A-D	:	-5 till +5 V -10 till +10 V
Mätområde Max	:	-10 till +10 V fullt utslag (1 ggr. först.)
Min	:	-39 till +39 mV fullt utslag (128 ggr först.)
Högsta upplösning	:	100 µV

==== DIFFERENTIALFÖRSTÄRKARE ===

Fast förstärkning	:	1 10 100
Motstånd Rg Kohm	:	100 10 1
Temperaturdrift	:	20 µV/grad Celsius
Offsetinställning	möjlig med potentiometer	

==== PROGRAMMERBAR FÖRSTÄRKARE (OPTION) ===

Förstärknig	:	1 2 4 8 16 32 64 128
-------------	---	---

==== A-D-OMVANDLARE ===

Antal bitar/nivåer	:	12 bitar / 4096 nivåer
Metod	:	Succesiv approximation
Mätområde, byglas	:	0 - 10 V, +- 5 V, +- 10 V
Omvandlingstid	:	ca 30 µS
Spänningsreferens	:	Intern i kretsen
Linjäritet	:	+- 1 bit
Temperaturdrift	:	+- 2 bitar vid min till max temperaturområde

==== KOMMANDON I BASIC ===

OUT 1,C	:	Kortval
OUT 4,O	:	'Single'-mätning
OUT 3,O	:	'Diff.'-mätning
OUT 2,K	:	K=Kanalval
OUT 0,O	:	Start konvertering av data
INP(1)	:	Bit 7 = "0" om klar, "1" om ej klar
	-	: Bit 4 = Extern bit från analog in kontakt
	-	: Bit 3-0 = Data bit 11-8
INP(0)	:	Data bit 7-0,
	-	: ställer om till ej klar (se INP(1) bit 7)
OUT 5,F	:	Förstärkning

ABC ADC32 anpassar olika typer av analoga spänningar till ABC 80. Detta sker i sex steg med analoga och digitala kretsar.

1. ==ANALOG MULTIPLEXER==

Kanalval styrs från ABC 80 med två analoga multiplexerkretsar på 16 kanaler var (HI-506A) med överspänningsskydd. Därpå följer en 4 kanalers multiplexer (LF 13333) för val av single eller differentiell mätning. Dessa ger tillsammans 32 analoga ingångar vid single-mätning och 16 vid differentiell mätning.

2. ==DIFFERENTIALFÖRSTÄRKARE==

Differentialförstärkaren har fast förstärkning på 1, 10 eller 100 gånger beroende på ett förstärkningsmotstånd (R_g). Offsetinställning kan ske med potentiometer R16.

3. ==PROGRAMMERBAR FÖRSTÄRKARE==

Med den programmerbara förstärkaren (MN 2020) är en option som man från ABC 80 kan styra förstärkningen in till AD-omvandlaren från 1 till 128 gånger. Denna styrning sker samtidigt med kanalvalet. När programmerbar förstärkare finns, bör differentialförstärkaren ha en fast förstärkning på 1. Offsetinställning kan ske med potentiometer R17.

4. ==SAMPLE/HOLD==

Sample/Hold-kresten (AD 582) håller den analoga signalen stilla under AD-omvandlarens konvertering. S/H styrs av AD-omvandlarens Hold-signal som är aktiv under konverteringen. Offsetinställning kan ske med potentiometer R15.

5. ==AD-OMVANDLARE==

AD-omvandlaren (AD 574) konverterar den analoga signalen till motsvarande digital nivå, på kommando från ABC 80. 12 bitar ger AD-omvandlaren, vilket ger 4096 nivåer. Den arbetar enligt metoden för succesiv appriximation och har en omvandlingstid på ca. 30 us. AD-omvandlaren kan kopplas bipolärt för +- 10V eller +- 5V inspänning eller kopplas unipolärt för 0V - 10V inspänning. Offsetinställning kan ske med potentiometer R14 och förstärkningstrimning kan ske med R12.

6. ==KONTROLLOGIK==

I kontrollogik ingår kortadresslogik, kommandoavkodningslogik och buffrar för AD-omvandlaren.

Först aktiveras AD-kortet med rätt kortadress. Sedan styrs det med de olika kommandona för kanalinställning, förstärkning, single eller differentiell mätning, start konvertering och avläsning av data.

Dessutom ingår DC/DC-omvandlare för +- 15V till de analoga kretsarna.

Alla analoga blocken kan ha offset-inställning men det är lämpligt att enbart ställa in detta på ett block.

En digital bit in finns tillgänglig på ingångssidan som kan användas för styrning eller kontroll i programmet.

Anpassningen av en analog signal utförs på två nivåer:

1. Anslutning av analog signal till ABC ADC32:s analoga input-sida.
2. Programmering av drivrutin för denna anslutning.

Nedanstående tabell visar stiftkonfigurationen på anpassningskontakten. De analoga ingångarna finns på C-stiften i nummerordning.

Anpassningkontakten på ABC ADC32 är ett Europadon 96/64-stifts kretskortsdon, vilket innebär att A- och C-raden är försedda med stift.

KANAL SINGLE DIFFERENTIELL

Nr	HÖG - LÅG	HÖG - LÅG
	Stift-Stift	Stift-Stift
0	C01 - A04	C01 - C17
1	C02 - A04	C02 - C18
2	C03 - A04	C03 - C19
3	C04 - A04	C04 - C20
4	C05 - A05	C05 - C21
5	C06 - A06	C06 - C22
6	C07 - A07	C07 - C23
7	C08 - A08	C08 - C24
8	C09 - A09	C09 - C25
9	C10 - A10	C10 - C26
10	C11 - A11	C11 - C27
11	C12 - A12	C12 - C28
12	C13 - A13	C13 - C29
13	C14 - A14	C14 - C30
14	C15 - A15	C15 - C31
15	C16 - A16	C16 - C32
16	C17 - A17	
17	C18 - A18	
18	C19 - A19	
19	C20 - A20	
20	C21 - A21	
21	C22 - A22	
22	C23 - A23	
23	C24 - A24	
24	C25 - A25	
25	C26 - A26	
26	C27 - A27	
27	C28 - A28	
28	C29 - A29	
29	C30 - A30	
30	C31 - A30	
31	C32 - A30	

GND A4 till A30 A4 till A30 Gemensam analog jord.
Bit in A03 A03

Observera att stiftsnumret är en siffra högre än motsvarande kanalnummer.

Vid differentiell mätning används bara kanalnummer 0 till 15. Förutom hög och låg signal skall man vid differentiell mätning även ansluta GND till signaljord. Hög och låg signal får ej avvika mer än 10V från signaljord.

Den digitala biten in (stift A03) avläses på bit 4 vid INP(1).

Kortval utförs med 6 byglingar i position 12 (se placeringsschema). ABC ADC32 rekommenderas att i första hand ha kortadress 17 och 18. Byglingarna är märkta med respektive vikt för kortvalet. Genom att addera ihop de vikter som ej har bygel, får man kortadressen. Om vikt 16 och 1 är uppklippta, får man kortadress 17. Med kommandot OUT 1,K aktiverar man kortet. K är kortadressen. Lysdioden visar om kortet är adresserat, och därmed aktivt. För ovanstående exempel kommer kortet att bli aktivt när man skriver OUT 1,17.

***** FÖRSTÄRKNING *****

--FAST FÖRSTÄRKNING--

Differentialförstärkarens fasta förstärkning ställs in med motståndet Rg. Detta är placerat mellan position 3 och 4. Följande förstärkningar kan fås:

$$\begin{array}{lll} Rg \text{ (Kohm)} & = 100 & 10 & 1 \\ \text{Förstärkning} & = & 1 & 10 & 100 \end{array}$$

--PROGRAMMERBAR FÖRSTÄRKARE--

När man monterar på en programmerbar förstärkare (MN2020) skall man på undersidan av kortet bryta upp förbindelsen mellan stift 7 och 10, markerat med en pil. Mellan position 5 och 7 finns plats för en trimpotentiometer (R17). Om man ej skall ha offsetinställning, monteras här ett motstånd på 10 Kohm mellan stift 8 och 9 på den programmerbara förstärkaren. Om man skall ha offsetinställning monteras här en trimpotentiometer på 10 Kohm.

***** TIDER *****

Följande inställningstider redovisas för de analoga komponenterna:

Diff. (AD521)	:Först.= 1 ger 7 us
	:Först.= 10 ger 5 us
	:Först.=100 ger 10 us

Prog.först.(MN2020):

Insvängningstid	:Först.= 1	2	4	8	16	32	64	128	
för 20 V sving	: us	= 2.5	3	4	6	8	17	33	65
till +- 0.1%	:								

Sample/hold (AD582): 3V/us

AD-omvandl. (AD574): Konverteringstid : ca 30 us

Omkring AD-omvandlaren finns en del byglingar som används för att ställa in mätområde, offset och förstärkningskontroll på denna.

Nedanstående visar hur man kopplar för en bipolär koppling.

== +-10V, +-5V UTAN OFFSET ==

Här visas hur man kopplar för att få en bipolär inkoppling utan offset och förstärkningskontroll.



Bygling för 20V sving väljs om man skall ha +-10V.

Bygling för 10V sving väljs om man skall ha +-5V.

R11 skall ej vara inkopplad

R12 = 50 ohm 0.5%

R14 = 50 ohm 0.5%





== +-10V, +-5V MED OFFSET ==

Denna koppling visar hur man kopplar för att få en bipolär inkoppling med offset och förstärkningskontroll.

Bygling för inspänning är enligt föregående inkoppling.

R12 = Trimpot. 89P 100 ohm (Först.)

R14 = Trimpot. 89P 100 ohm (Offset)



Mät område ± 5 V

Nedanstående visar hur unipolär inkoppling utförs.

==OV - 10V UTAN OFFSET==



Här visas hur man kopplar för att få en unipolär inkoppling utan offset och förstärkningskontroll.

Bygling för 10V sving väljs när man har OV - 10V.
ABC ADC levereras normalt med denna koppling.

R14 = 50 ohm 0.5%

R12 = 50 ohm 0.5%

R13 = (50 ohm 0.5%), används ej



==OV - 10V MED OFFSET==

Denna koppling visar hur man kopplar för att få en unipolär inkoppling med offset och förstärkningskontroll.



Bygling för 10V sving väljs när man har OV - 10V.

R11 = 50 ohm 0.5%

R12 = Trimpot. 89P 100 ohm (Först.)

R13 = 100 Kohm

R16 = Trimpot. 100kohm (Offset)



Från ABC 80 styrs och läses ABC ADC32 med olika kommandon.
Följande kommandon finns i BASIC:

OUT 0,X	:Start av konvertering på ADC
OUT 1,C	:Kortval för aktivering av kort med adress C
OUT 2,K+32xF	:Inställning, K=Kanalval, 2xxF=Förstärkning
OUT 3,X	:Differentiell mätning, 16 kanaler
OUT 4,X	:Single mätning, 32 kanaler
INP(0)	:Inläsning av data bitar 0 - 7 från ADC
INP(1)	:Bit 7 = '1' ej klar, '0' klar :Bit 4 = Digital ingång från ingångsstift A03 :Bit 0 - 3 = Inläsning av data bitar 8 - 11 från ADC

Ett drivprogram för inläsning från ABC ADC32 skall utföra följande steg (uppgifter inom parantes gäller vid programmering i assembler):

1. Gör kortval för att aktivera kortet, OUT 1,16
2. (Ställ kortet i 'ej klar' med INP(0))
3. Välj Diff. eller Single-mätning, OUT 3,0 alt. OUT 4,0
4. Välj kanal (K) och ev. förstärkningsfaktor (F), OUT 2,K+32xF
5. (Vänta för att förstärkarna skall stabilisera sig)
6. Starta omvandlingen, OUT 0,0
(Efter start kan nästa analoga kanal ställas in med multiplexern, eftersom den just akutella kanalen finns lagrad i Sample/hold-kretsen.)
7. (Vänta på att omvandlingen är klar genom att läsa status på INP(1), bit 7 tills den blir '0')
8. Läs av data, de fyra högsta bitarna,(bit 8-12) A=(INP(1) AND 15)
9. Läs av data, de åtta lägre bitarna, B=INP(0)
10. Sätt samman mätvärdet, C=Ax256+B
11. Gå till huvudprogram eller repetera avläsningen från punkt 4 (alternativt punkt 5 om nästa kanal redan är inställt) om fler kanaler skall läsas av.

Nedanstående exempel visar hur ett enkelt program ser ut för avläsning av samtliga kanaler på ADC ADC32.

```

10 REM ***** ADCLÄS1 *****
20 REM DATUM HMSCM, Skriver alla kanalers
30 REM spänning vid 0 - 10 V koppling, single mätning
40 REM -----
50 PRINT CHR$(12); "MÄTNING MED ABC ADC32 PÅ ALLA KANALER"
60 PRINT "Nr Spänning" Nr Spänning"
70 R%=4% : REM Rad
80 C%=0% : REM Kolumn
90 REM =====INIT-ADC=====
100 OUT 1%,16% : REM Kortval
110 OUT 4%,0% : REM Single mätning
120 REM =====MÄT-ADC=====
130 FOR K%=0% TO 31% : REM Kanalval
140 OUT 2%,K% : REM Kanalval
150 OUT 0%,0% : REM Start konvertering
160 V2%=(INP(1%) AND 15%) : REM Data bit 8 - 11
170 V1%=INP(0%) : REM Data in bit 0 - 7
180 V%=V1%+V2%*256% : REM Beräkna nivå
190 V=V%*10/4095 : REM Beräkna spänning
200 REM =====SKRIV DATA=====
210 IF R%=20% THEN R%=4% : C%=20% : REM Ny spalt ?
220 PRINT CUR(R%,C%)K%;V : REM Skriv data
230 R%=R%+1%
240 NEXT K%
250 GOTO 70

```

Nedanstående program visar hur man någorlunda snabbt i BASIC läser av en kanal, lägger upp den i en vektor och i efterhand skriver ut mätresultatet.

```

10 REM ***** ADCLÄS2 *****
20 REM DATUM HMSCM, Utskrift av en kanals nivå och
30 REM spänning vid 0 - 10V koppling, Diff. mätning
40 REM -----
50 PRINT CHR$(12%); "ANALOGMÄTNING MED ABC ADC32"
60 PRINT "Vilken kanal skall mätas ";
70 INPUT K%
80 GOSUB 190 : REM Initiera ABC ADC
90 PRINT "Hur många mätningar ";
100 INPUT I%
110 DIM V%(I%) : REM Nivåvektor
120 PRINT "Nummer" Nivå Spänning"
130 GOSUB 230 : REM Mät
140 FOR J%=1% TO I% : REM Nummer
150 V=V%(J%)*10/4095 : REM Beräkna spänning
160 PRINT J%,V%(J%),V : REM Skriv värden
170 NEXT J%
180 END
190 REM *** ADC-INIT-KANAL-SUBRUTIN ***
200 REM Kortval, Singlemätning & Kanalval
210 OUT 1%,16%,3%,0%,2%,K%
220 RETURN
230 REM ***** ADC-MÄT-SUB-RUTIN *****
240 REM INDATA I%=Antal mätningar, UTDATA V%(I%)=Nivåvektor
250 FOR J%=1% TO I%
260 OUT 0%,0% : REM Start konvertering
270 V%(J%)=SWAP%(INP(1%) AND 15%)+INP(0%) : REM Indata från ADC
280 NEXT J%
290 RETURN

```

Nedanstående program visar hur ett assemblerat program kan se ut.
 I detta fall är det placerat i POKE-arean längst upp i minnet.
 Man använder programmet genom att göra CALL (-128%, K%), där K% är
 kanalnummer. Om K% är positivt utförs singlemätning och om K% är
 negativt, utförs differentiell mätning.

ABC80-ASSEMBLER

SIDA 1

```

0000 ;***** ADCLÄS *****
0000 ; VERSION 1.0 DATUM SIGN
0000 ; AVLÄSNINGRUTIN I ASSEMBLER AV ABC ADC32
0000 ;
0000 ; INPUT DE = Kanalnr + mätmode
0000 ; Singlemätning om Kanalnr är positivt.
0000 ; Diff.mätning om Kanalnr är negativt.
0000 ; OUTPUT HL = Nivå på avläst kanal
0000 ; DELAY-tiden beror på
0000 ; förstärkning i diff.förstärkaren
0000 ;
0000 DATA EQU 0 ;Import data
000A DELAY EQU 10 ;Tidsfördröjning
0003 DIFF EQU 3 ;Utport Diff.mätning
0010 KADR EQU 16
0001 KVAL EQU 1 ;Utport kortval
0002 MUXSET EQU 2 ;Utport Kanalinställning
0004 SINGLE EQU 4 ;Utport Singlemätning
0000 START EQU 0 ;Utport start omvandling
0001 STATUS EQU 1 ;Import Status
0000 ;
FF80 ORG 65408
FF80 3E10 ADLÄS LD A,KADR ;Aktivera kort
FF82 D301 OUT (KVAL),A
FF84 DBO0 IN A,(DATA) ;Återställ till ej klar
FF86 OE04 LD C,SINGLE ;
FF88 7B LD A,E ;Kanal
FF89 CB7A BIT 7,D ;Negativt ?
FF8B 2804 JR Z,SDSET ; Nej
FF8D ED44 NEG ;Vänd rätt på kanal
FF8F OE03 LD C,DIFF
FF91 ED79 SDSET OUT (C),A ;Sätt single eller diff.
FF93 D302 OUT (MUXSET),A;Sätt kanal
FF95 3EOA LD A,DELAY ;Vänta tills förstärkarna
FF97 3D WAIT DEC A ; är stabila
FF98 20FD JR NZ,WAIT ;Hopp om ej klar
FF9A D300 OUT (START),A ;Starta omvandling
FF9C DB01 LOOP IN A,(STATUS);Vänta på omvandling klar
FF9E CB7F BIT 7,A ;
FFAO 20FA JR NZ,LOOP ;Ej klar
FFA2 E60F AND OFH ;4 högre bitar
FFA4 67 LD H,A ; till högre delen
FFA5 DBO0 IN A,(DATA) ;8 lägre bitar
FFA7 6F LD L,A ; till lägre delen
FFA8 C9 RET ;Tillbaka till BASIC
0000 END

```


ABC80-ASSEMBLER

SIDA 2

ADLÄS	FF80	DATA	0000
DELAY	000A	DIFF	0003
KADR	0010	KVAL	0001
LOOP	FF9C	MUXSET	0002
SDSET	FF91	SINGLE	0004
START	0000	STATUS	0001
WAIT	FF97		


Nedanstående exempel visar hur den analoga signalen kan anslutas till analogingångarna på ABC ADC32

EXEMPEL 1. Enkel analog spänning på kanal 0.



Spänningen kan vara 10V, +-5V, +-10V fullt utslag, alternativt 1/10 eller 1/100 om man har fast förstärkning.


EXEMPEL 2. Differentiell analog spänning på kanal 1.



Denna koppling är lämplig vid låga spänningar, när man använder fast förstärkning högre än 1 eller använder den programmerbara förstärkaren.


Observera att man måste koppla ihop de analoga jordarna, för differentialförstärkaren mäter båda insignalerna relativt analog jord.

EXEMPEL 3. Differentiell analog spänning med enkelt mätobjekt på kanal 4.




Denna koppling används när man inte har en differentiell mätsignal, men vill ändå mäta differentiellt på grund av andra störningar.

EXEMPEL 4. Differentiell analog spänning på kanal 6 i fall med separata jordar.



När de båda analoga jordarna ej direkt kan sammankopplas, kan man använda denna metod under förutsättning att signalerna ej avviker mer än 10V. R dimentioneras så att mätsignalen ej påverkas.



I denna koppling mäts en ström, exv 4 - 20 mA, med ett shuntmotstånd R_s , som har en känd resistans. 50 ohm ger vid 20 mA en spänning på 1V. Man ställer då in den fasta förstärkningen på 10 gånger och området 0-10V, vilket ger fullt utslag vid 1V.
 R_j används för att få spänningen över R_s refererad till jord och bör väljas så högt att ingen nämnvärd ström flyter denna väg tillbaka till strömobjektet.

***** METODER FÖR STÖRUNDERTRYCKNING *****

Vid anslutning av analoga signaler måste man ta hänsyn till en del faktorer som kan påverka mätningen i den analoga delen av ABC ADC32.

1. En gemensam jordpunkt bör användas för att eliminera jordströmmar som kan påverka den analoga signalen.
2. Yttre störande signaler avskärmas genom att använda skärmad kabel mellan mätobjekt och ABC ADC32.
3. Låga spänningar förstärks upp i förförstärkare innan de kommer in till analogingången på ABC ADC32.
4. De analoga ledningarna skall vara skilda från ABC80-s busskabel.
5. Man kan i program bilda medelvärde av flera mätningar, under förslagsvis 20 ms för att eliminera nätbrum.
6. Vid hög förstärkning på ABC ADC32 finns risk att DC/DC-omvandlaren stör de analoga komponenterna. Man kan då koppla bort denna och ersätta den med +- 12V-spänning från bussen. Detta innebär att man inte kan utnyttja fullt utslag på de analoga komponenterna utan ungefär 75% av fullt utslag.

Till ABC ADC32 finns en del drivprogram för att enklare och snabbare hantera analoga signaler i BASIC. Drivprogrammen ger:

- Snabba mätningar med kanalbyte.
- Mätningar med exakt tidsintervall i multiplar av 20 ms eller 1 s.
- externt stydda mätförlopp.
- Lättåtkomligt resultat.
- Enkel hantering från BASIC.
- Laddning av en matris med mätvärden.

Drivrutinerna är gjorda för olika applikationer, men en och samma drivrutin kan användas på flera olika sätt genom det optionsval som anges när man öppnar rutinen.

Drivrutinen läses in före huvudprogrammet varvid den reserverar ett eget utrymme i RAM-minnet.

Följande rutin kallas ADF: och används vid snabba avläsningar av analoga signaler.

```

10 REM ***** ADFPROV1 *****HMSCM*****
20 REM SNABB LÄSNING AV AV KANAL 0-4
30 REM 20 MÄTVÄRDEN
40 REM 10V, EXTERN STYRNING, DIFF.MÄTNING
50 OPEN "ADF:AD1" ASFILE 1 : REM ÖPPNING MED OPTIONER
60 DIM M%(4%,20%) : REM MATRIS FÖR INLAGRING AV MÄTVÄRDEN
70 PRINT #1,"M%" : REM MATRIS TILL DRIVRUTIN
80 M%(1%,0%)=1% : REM KANAL 1
90 M%(2%,0%)=2% : REM KANAL 2
100 M%(3%,0%)=3% : REM KANAL 3
110 M%(4%,0%)=4% : REM KANAL 4
120 INPUT #1,A$ : REM UTFÖR MÄTNING
130 REM ----- UTSKRIFT AV DATA -----
140 PRINT "KANAL 1 KANAL 2 KANAL 3 KANAL 4"
150 FOR N%=1% TO 20%
160 FOR K%=1% TO 4%
170 PRINT M%(K%,N%),
180 PRINT TAB (K*9%);
190 NEXT K%
200 PRINT
210 NEXT N%
220 END

```

I rad 50 öppnas kommunikationen till drivrutinen med option AD1.
A innebär kortadress 17.

D innebär extern styrning, 10V differentiell mätning.

1 innebär fast förstärkningsfaktor på 1.

Dessa optioner kan man läsa ut av beskrivningen till denna drivrutin.

I rad 70 definieras att matrisen M% skall fyllas med mätvärden.

I rad 120 fylls matrisen med mätvärden i den takt som bestäms från den digitala ingången.

1. Anslut de analoga signalerna till ett kontaktdon som passar i den analoga ingångskontakten på ABC ADC32. Se sidan 4 för stifts placering.
2. Bygla ABC ADC32 för spänningssområde, förstärkning och kortadress. Se sidan 5 för kortadress, och sidan 6 och 7 för andra optioner.
3. Stäng av alla ingående enheter som skall kopplas in.
4. Sätt ABC ADC32 på en kortplats i FD2, FD2U, EX1 eller EX8.
5. Koppla in det analoga kontaktdonet till den analoga ingångskontakten.
6. Sätt på strömmen på alla ingående enheter.
7. Testa om kortvalet är riktigt med följande kommando:
OUT 1,C där C är kortadressen.
Lysdioden skall då tändas.
8. Testa de analoga ingångarna med ett BASIC-program av samma typ som ADCLÄS1 och ADCLÄS2 på sidan 9.
9. Testa den digitala ingången på stift AO3 med följande program:

```
10 REM ***** ADCDIG *****
20 REM LÄS DIGITAL INGÅNG PÅ ABC ADC32
30 OUT 1%,17% : REM KORTVAL
40 IF (INP(1%) AND 16%)=16% THEN PRINT 1 ELSE PRINT 0
50 GOTO 40
```

Om den digitala ingången är logisk etta skrivs 1 och om den är låg skrivs 0.

===== UNIPOLÄR KALIBRERING =====

Vid kalibrering används kanal 0 med 10 volts sving.
Ett enkelt program används som bara skriver mätvärdet på kanal 0:

```
10 **** ADCTRIM ****
20 OUT 1%,17%,3%,0%,2%,0% : REM kort 17,single, kanal 0
30 OUT 0%,0% : REM Start ADC
40 PRINT SWAP%(INP(1%) AND 15%)+INP(0%)
50 GOTO 30
```


Trimningen utförs på följande sätt:

1. Anslut en spänning på 1.22 mV till kanal 0 (0.5 bit från MIN).
2. Starta programmet och trimma in offset med R16 (alt. R14, R15 el. R17) så att mätvärdet slår mellan 0 och 1 för denna spänning.
3. Anslut 9.9963 V till kanal 0 (1.5 bit under MAX).
4. Programmet skall nu visa ett mätvärde på 4094 till 4095.
5. Används R12 (GAIN TRIM) kan man här trimma in så att mätvärdet slår mellan 4094 och 4095.

===== BIPOLÄR KALIBRERING =====

Den bipolära kalibreringen utförs på samma sätt som den unipolära.

1. Anslut -4.9988 V till kanal 0 (0.5 bit från MIN).
2. Trimma offset med R16 (alt. R14, R15 el. R17) så att mätvärdet slår mellan 0 och 1.
3. Anslut +4.9963 V till kanal 0 (1.5 bit under MAX).
4. Programmet skall nu visa ett mätvärde på 4094 till 4095.
5. Används R12 (GAIN TRIM) trimmas mätvärdet in mellan 4094 och 4095.



KOMPONENTFÖRTEKNING

POS	KRET
1	74LS244
2	74LS367
3	DM8131
4	74LS107
5	74LS367
6	Kortvalsbytning
7	74LS174
8	S/H AD582K
9	ADC AD574J
10	(Prog. först. BB3606)
11	DC/DC-konv. USR15-15
12	Diff. först. AD521
13	MUX LF13333
14-15	MUX HI506-a eller IH6116
C1-C5	6,8 μ F 6,3V
C6-C7	22 μ F 6,3V
C8-C12	22 μ F 16V
C13	33 μ F 16V
C14-C15	0,1uF
C16	1 nF
C17	150 Pf (1nF vid BB-3606)
C18	560 Pf
C19, C20	0,1 μ F

POS	KRET
R9	100 kohm 0,5% vid F=1 10 kohm 0,5% vid F=10 1 kohm 0,5% vid F=100
R1-R2	1 kohm 5%
R3	100 kohm 0,5%
R4-R6	1 kohm 5%
R7	2 ohm 5%
R8	300 ohm 5%
R9-R10	7x1 kohm
R11	50 ohm 5% vid 0-10 V
R12	50 ohm, utan trim GAIN 89P100 ohm trim GAIN
R13	50 ohm utan offset ADC se vidare sid 6,7.
R14	50 ohm utan offset ADC
R15	89P10 Kohm offset S/H
R16	89P10 Kohm offset ADC
R17	89P10 Kohm offset Diff eller Prog. först.
R18	89P100 kohm offset Prog. först.
D1	Lysdiod

Option. När Progrämerbar förstärkare används:
Diff. först. AD 521 byttes till Prog. först BB3606 och
offset trim R18 monteras.